
Example nº 4
Earth retaining wall

CivilFEM Manual of Advanced Examples

Example nº 4 – Table of Contents

- 4 EXAMPLE Nº 4: EARTH RETAINING WALLS..... 1
 - 4.1 AIM 1
 - 4.2 DESCRIPTION OF THE EXAMPLE 1
 - 4.3 RESULTS TO OBTAIN..... 3
 - 4.4 CALCULATION LOG 3
 - 4.4.1 Introduction 3
 - 4.4.2 Preprocess..... 4
 - 4.4.3 Solution..... 5
 - 4.4.4 Postprocess 5
 - 4.5 RESULTS 6
 - 4.6 SUMMARY 8

4 EXAMPLE N° 4: EARTH RETAINING WALLS

4.1 AIM


The aim of this example is to show the calculation tools for earth retaining walls built in layered terrains.

4.2 DESCRIPTION OF THE EXAMPLE

The wall considered must retain two terrains and it is embedded in a granitic rock base.

In the following figure the geometry and the characteristics of the materials that form the terrain and the wall are show.

The excavation will be done in three stages, notated as *Step1*, *Step2* and *Step3*. At the end of each one of the two first stages, wall anchorages are set. These anchorages can be considered as fixed supports.



DATA:

GEOMETRY

The level, in meters referred to the sea level are:

z1	=	925.14	ze1	=	921.00
z2	=	919.20	ze2	=	915.00
z3	=	913.00	ze3	=	909.00
z4	=	890.00	zrw	=	900.00

WALL

The wall has width of $e_{rw} = 0.80$ m.

A C25/30 concrete (EC2) will be used for the wall and a S400 steel (EC2) for the passive reinforcement that will be placed symmetrically.

TERRAIN

The material properties of the terrains are not known, so the mean values of CivilFEM will be used.

4.3 RESULTS TO OBTAIN

The results that will be obtained are the bending moments law at the end of each excavation phase, and the envelope of all of them.

The wall reinforcement will be later designed, according to the European concrete code (Eurocode 2), for the axial force and bending moment.

4.4 CALCULATION LOG**4.4.1 Introduction**

Like in other models, it starts with the selection of the code (Eurocode 2) that will be used for checking and design, and the units system that will be used (International System).

Next, the parameters that define the geometry of the model are established.

In order to be able to build the wall, it is necessary to define the composition of the layered terrain. To do so, CivilFEM materials are selected from the library, not being necessary, in this case, to change any of the properties already defined. The concrete and reinforcement steel materials that they will be used in the wall sections are selected.

The following step is to group the geotechnical materials in the terrain, defining the thickness of each one of the layers.

It is also necessary to define the rest of properties of the beams that will build the vertical wall: cross section, beam properties and type of element. A symmetrical reinforcement of $1 \text{ cm}^2/\text{m}$ is set at each face so that CivilFEM, in postprocessing phase, will design.

Finally the wall is defined, establishing its dimensions, the type of element to use, beam properties, height of excavation and location and calculation in which the anchorages will be placed.

It is important to consider that the number of calculation steps does not have to be the same as the number of excavation phases. It may be possible to have calculation steps in which supports or anchorages are modified, maintaining the excavation height at the same level.

In order to improve the convergence of the solution, excavation phases between the ones stated in the description of the example have been set.

Once all the needed data to build the wall has been defined, the model is generated, with an element maximum size of 0.25 m.

The solution phase for this model is very simple. Firstly the maximum number of equilibrium iterations is set, since the default number may be small for this type of calculation. Next all the phases of the constructive process are solved.

During the execution it is possible that floating windows appear to confirm the solve of each one of the phases. This is due to the fact that CivilFEM changes the properties of the materials and elements in each step, to fit the values of the earth pressures. As it is a non standard use, it requests confirmation in order to continue with the following load step. These windows do not appear if the model is launched in *BATCH* mode.

Once the solving has ended, the bending moments are plotted for each one of the excavation phases specified in the description of the example, and the envelope of these bending moments laws is obtained. In order to obtain the envelope, a results combination is made performed This combination is made up of incompatible states (two excavation stages cannot happen simultaneously). The targets for the combination are the maximum and minimum bending moments.

It is important to emphasize that the chosen combination target searches for the results in the cross sections (`~TRGDEF,1,CROSS,M,Z,MIN`) and not in the beam elements (`~TRGDEF,1,BEAM,M,Z,MIN`). The results that appear in the combined states are the ones inside the same results group of the selected target and therefore, in order to plot the bending moments using CivilFEM command, it is necessary to obtain results from the cross sections.

Finally the bending reinforcement is designed according to Eurocode 2, for each one of the calculation steps, and the envelope of these results is obtained.

4.4.2 Preprocess

```
FINISH
~CFCLEAR,,1
~CFACTIV,GETC,Y

~CODESEL,,EC2
~UNITS,SI

/PREP7
! Parameters
! Terrain heights
z1 = 925.14 ! Terrain top height.
z2 = 919.20 ! Contact between layers 1 and 2.
z3 = 913.00 ! Contact between layers 2 and 3.
z4 = 890.00 ! Terrain bottom level.
! Specified Excavation heights
zel = 921.00 ! Excavation step 1.
```

```

ze2 = 915.00 ! Excavation step 2.
ze3 = 909.00 ! Excavation step 3.
! Calculation excavation heights.
N_Height = 6
*DIM,Height,ARRAY,N_Height
Height( 1) = 923.00
Height( 2) = ze1      $ Sup1_Step = 2 + 1 $ Exc1_Step = 2
Height( 3) = 916.00
Height( 4) = ze2      $ Sup2_Step = 4 + 1 $ Exc2_Step = 4
Height( 5) = 911.00
Height( 6) = ze3      $ Exc3_Step = 6
! Wall
zrw = 900.00 ! Wall bottom level
erw = 0.8    ! Wall thickness

! Material definition
~CFMP,1,LIB,CONCRETE,EC2,C25/30
~CFMP,2,LIB,REINF,EC2,S400
~CFMP,3,LIB,SOIL,,SW
~CFMP,4,LIB,SOIL,,CL
~CFMP,5,LIB,ROCK,,Granite

! Terrain definition
~TERDEF,1,NEW,3,0,Y,z1,0,0,0
~TERDEF,1,LAYER,1,3,z1-z2,0,DB
~TERDEF,1,LAYER,2,4,z2-z3,0,DB
~TERDEF,1,LAYER,3,5,z3-z4,0,DB

! Wall Cross section and Beam Properties
~CSECDMS,1,REC,1,erw,1
~RNFDEF,1,1,2,1,0,0.04,1E-4
~RNFDEF,1,2,2,3,0,0.04,1E-4
~BMSHPRO,1,BEAM,1,1,,,54,1,0,

! Element type for wall elements
ET,1,BEAM54

! Wall and excavation process definition.
~WALLINI,0,N_Height,0,0,1,1,z1,1
~WALLGEN,1,1,z1-zrw,1,1,1
*DO,II,1,N_Height
  ~WALLSTP,II,1,Height(II)
*ENDDO
~WALLANC,Sup1_Step,1,ze1,2
~WALLANC,Sup2_Step,1,ze2,2

! Model
~WALLMOD,0.25 ! Maximum element size = 0.25 m

```

4.4.3 Solution

```

/SOLU
! Convergence parameters
NEQIT,100
! Solve the model
~WALLSOL,ALL

```

4.4.4 Postprocess

```

/POST1
! Plot bending moments for each excavation step.

```

EXAMPLE N° 4: EARTH RETAINING WALLS

```
~CFSET,,Exc1_Step
~PLLSFOR,M,Z
~CFSET,,Exc2_Step
~PLLSFOR,M,Z
~CFSET,,Exc3_Step
~PLLSFOR,M,Z

! Combinations
~CMBCLR,,1
~TRGDEF,1,CROSS,M,Z,MIN
~TRGDEF,2,CROSS,M,Z,MAX
/TITLE,Excavation steps
~CMBDEF,1,INCOMPAT,3,0,1
~STSTDEF,1,1,LSTEP,Exc1_Step,0,1,1,1,1
~STSTDEF,1,2,LSTEP,Exc2_Step,0,1,1,1,1
~STSTDEF,1,3,LSTEP,Exc3_Step,0,1,1,1,1
~COMBINE


! Plot maximum and minimum bending moments.
ESEL,S,MAT,,1 ! Select concrete elements only (wall)
~CMBDAT,2 ! Point to combined results
~CFSET,,1,1
~PLLSFOR,M,Z
~CFSET,,1,2
~PLLSFOR,M,Z

! Concrete design
~CMBDAT,1 ! Point to original results
*DO,II,1,N_Height
  ~CFSET,,II
  ~DIMCON,2DB, ,0,1,100
*ENDDO
! Reinforcement amount envelope
~ENVDEF,1,N_Height,1
~ENVELOP,MAX ! This will create a new alternative (N_Height+1)


! Plot results of the envelope
~cfset,N_Height+1
~PLLSCON,REINFACT,1
```

4.5 RESULTS

The obtained maximum and minimum bending moments law are the following ones:



And the envelope of the reinforcement amount for all the calculation steps is:



It is important not to forget that the obtained values are the coefficients that must be used to scale the initial reinforcement, so that it complies with Eurocode 2. This is to say, an initial reinforcement $1 \text{ cm}^2/\text{m}$ was placed in each face, and the obtained result is $\omega_{\max} = 19.9$. This means that strict reinforcement amount needed is $19.9 \text{ cm}^2/\text{m} \approx 20 \text{ cm}^2/\text{m}$ at the most critical point.

4.6 SUMMARY

With the resolution of this exercise the reader has been able to practice the way in which retaining walls with excavation process can be studied with CivilFEM. In particular, the following subjects have been practiced:

- Definition of multi-layered terrains.
- Retaining walls.
- Combinations and envelopes.
- Concrete reinforcement design.

Which generally include all the needs that may come out at the time of designing or verifying a retaining wall.